
How to Install and use the National Semiconductor 

LM741 OpAmp Subcircuit Model in LTspice Netlists 
 

Let’s suppose you want to simulate the circuit shown in Figure 1, with a realistic (non-ideal) 

model for the LM741 operational amplifier. 

 

 

 

 

 

 

 

 

 

 

 

Figure 1 

 

 

LTspice doesn’t include a model for the LM741operational amplifier, so I’ll start by doing an 

internet search for “LM741 SPICE model”.  Several possibilities are found.  Among them is a link 

to the Texas Instruments website for a model that was developed by National Semiconductor, as 

shown in Figure 2.  (National Semiconductor became the “Silicon Valley” division of Texas 

Instruments in 2011.)  On the TI site it’s called a PSpice model, but that doesn’t matter because 

LTspice and PSpice use the same structure and syntax for such models, as do virtually all SPICE-

based simulators. 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

Figure 2 

 

 



Clicking on this link takes you to http://www.ti.com/product/LM741/toolssoftware (See Figure 

3.), where you can download a compressed file named snom211.zip that contains the file we really 

want – LM741.MOD – a SPICE model file.  

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

Figure 3 

http://www.ti.com/product/LM741/toolssoftware


Download and extract/unzip the file.  Note that once it’s unzipped, it’s a SPICE model file with 

extension MOD (See Figure 4.). 

 

 

 

 

 

 

 

 

 

 

 

 

Figure 4 

 

 

Now you can examine its content – it’s a text file, so you can use Notepad – and you’ll see that the 

model is defined as a SPICE subcircuit called LM741/NS  (See Figure 5.). 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

Figure 5 

 

Since it’s a subcircuit description, copy the file LM741.MOD to the folder 

…\Documents\LTspiceXVII\lib\sub\  on your computer. 

 

Now you’ve made the LM741 opamp a permanent part of your LTspice component library, and 

you’re ready to create a netlist and do the simulation.  First, label everything as you normally 

would – my labels are shown in Figure 6 – and then write the netlist required for the simulation.  

Note how I have added two voltage sources so as to bias the opamp properly, but they are not 

shown connected directly to the device in order to make the diagram neater than might be otherwise 

expected.  The fact that the node numbers for the bias sources and the terminal connections on the 

opamp itself are the same means that LTspice will interpret them as being physically connected 

together. 



 

 

 

 

 

 

 

 

 

 

 

Figure 6 

 

My netlist is shown in Figure 7. 

 

 
* Q:\LTspice\My Stuff\LM741 OpAmp Inverter.cir 
.include LM741.MOD 
V+     4     0     DC     15      
V-     5     0     DC     -15 
X1     0     2     4      5     3     LM741/NS 
Vin    1     0     DC     1 
Rf     2     3     40k 
Rin    1     2     10k 
RL     3     0     100k 
.op 
.end 

 

Figure 7 

 

Note: 

1. Inclusion of the “.include LM741.MOD” directive to tell LTspice where to look for the 

LM741 subcircuit model, and 

2. The opamp requires five node connections, as shown in Figure 8. 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

Figure 8 



The corresponding output is shown in Figure 9. 

 

 
       --- Operating Point --- 
 
V(4):  15  voltage 
V(5):  -15  voltage 
V(2):  0.00102008  voltage 
V(3):  -3.99082  voltage 
V(1):  1  voltage 
I(Rl):  -3.99082e-05  device_current 
I(Rin):  9.9898e-05  device_current 
I(Rf):  9.97961e-05  device_current 
I(Vin):  -9.9898e-05  device_current 
I(V-):  0.00179686  device_current 
I(V+):  -0.001657  device_current 
Ix(1:1):  5.77868e-08  subckt_current 
Ix(1:2):  1.01929e-07  subckt_current 
Ix(1:99):  0.001657  subckt_current 
Ix(1:50):  -0.00179686  subckt_current 
Ix(1:28):  0.000139704  subckt_current 

 

Figure 9 

 

 

As mentioned earlier, now that you have the LM741 model installed, you’ll have it available to 

use in other simulations at any time in the future. 

 

This method of installation will work for virtually any device model you may download (or 

develop yourself).  Most manufacturers have such models available for download from their 

website. 

 

FYI, the complete content of the LM741.MOD file is shown in Figure 10. 

 

You might also find the LT-wiki Components Library and Circuits useful for other applications. 

 

 
*////////////////////////////////////////////////////////////////////// 

* (C) National Semiconductor, Inc. 

* Models developed and under copyright by: 

* National Semiconductor, Inc.   

 

*///////////////////////////////////////////////////////////////////// 

* Legal Notice: This material is intended for free software support. 

* The file may be copied, and distributed; however, reselling the  

*  material is illegal 

 

*//////////////////////////////////////////////////////////////////// 

* For ordering or technical information on these models, contact: 

* National Semiconductor's Customer Response Center 

*                 7:00 A.M.--7:00 P.M.  U.S. Central Time 

*                                (800) 272-9959 

* For Applications support, contact the Internet address: 

*  amps-apps@galaxy.nsc.com 

http://ltwiki.org/?title=Components_Library_and_Circuits


 

*////////////////////////////////////////////////////////// 

*LM741 OPERATIONAL AMPLIFIER MACRO-MODEL 

*////////////////////////////////////////////////////////// 

* 

* connections:      non-inverting input 

*                   |   inverting input 

*                   |   |   positive power supply 

*                   |   |   |   negative power supply 

*                   |   |   |   |   output 

*                   |   |   |   |   | 

*                   |   |   |   |   | 

.SUBCKT LM741/NS    1   2  99  50  28 

* 

*Features: 

*Improved performance over industry standards 

*Plug-in replacement for LM709,LM201,MC1439,748 

*Input and output overload protection 

* 

****************INPUT STAGE************** 

* 

IOS 2 1 20N 

*^Input offset current 

R1 1 3 250K 

R2 3 2 250K 

I1 4 50 100U 

R3 5 99 517 

R4 6 99 517 

Q1 5 2 4 QX 

Q2 6 7 4 QX 

*Fp2=2.55 MHz 

C4 5 6 60.3614P 

* 

***********COMMON MODE EFFECT*********** 

* 

I2 99 50 1.6MA 

*^Quiescent supply current 

EOS 7 1 POLY(1) 16 49 1E-3 1 

*Input offset voltage.^ 

R8 99 49 40K 

R9 49 50 40K 

* 

*********OUTPUT VOLTAGE LIMITING******** 

V2 99 8 1.63 

D1 9 8 DX 

D2 10 9 DX 

V3 10 50 1.63 

* 

**************SECOND STAGE************** 

* 

EH 99 98 99 49 1 

G1 98 9 5 6 2.1E-3 

*Fp1=5 Hz 

R5 98 9 95.493MEG 

C3 98 9 333.33P 

* 

***************POLE STAGE*************** 



* 

*Fp=30 MHz 

G3 98 15 9 49 1E-6 

R12 98 15 1MEG 

C5 98 15 5.3052E-15 

* 

*********COMMON-MODE ZERO STAGE********* 

* 

*Fpcm=300 Hz 

G4 98 16 3 49 3.1623E-8 

L2 98 17 530.5M 

R13 17 16 1K 

* 

**************OUTPUT STAGE************** 

* 

F6 50 99 POLY(1) V6 450U 1 

E1 99 23 99 15 1 

R16 24 23 25 

D5 26 24 DX 

V6 26 22 0.65V 

R17 23 25 25 

D6 25 27 DX 

V7 22 27 0.65V 

V5 22 21 0.18V 

D4 21 15 DX 

V4 20 22 0.18V 

D3 15 20 DX 

L3 22 28 100P 

RL3 22 28 100K 

* 

***************MODELS USED************** 

* 

.MODEL DX D(IS=1E-15) 

.MODEL QX NPN(BF=625) 

* 

.ENDS 

*$ 

 

Figure 10 

 


